Maple-assisted proof of formula for A231581

Robert Israel

6 December 2017

There are $4^4 = 256$ possible configurations for a 2×2 sub-array. Consider the 256×256 transition matrix T such that $T_{ij} = 1$ if the bottom two rows of a 3×2 sub-array could be in configuration i while the top two rows are in configuration j (i.e. the middle row is compatible with both i and j, and no entry in that row is less than three of its five neighbours), and 0 otherwise. The following Maple code computes it. I'm encoding a configuration

$$\begin{bmatrix} b_1 & b_2 \\ b_3 & b_4 \end{bmatrix}$$

as $b+1$ where $b_1 b_2 b_3 b_4$ is the base-4 representation of b. The $+1$ is needed because matrix indices start at 1 rather than 0.

```
> q:= proc(a,b) local r,s,t;
     s:= floor((a-1)/16);
     if s <> (b-1) mod 16 then return 0 fi;
     s:= convert(s+16,base,4);
     r:= convert(16+floor((b-1)/16),base,4);
     t:= convert(16+ ((a-1) mod 16),base,4);
     if nops(select(x -> x > s[1], [s[2],r[1],r[2],t[1],t[2]])) >=
  3 then return 0 fi;
     if nops(select(x -> x > s[2], [s[1],r[1],r[2],t[1],t[2]])) >=
  3 then return 0 fi;
     1
  end proc:
  T:= Matrix(256,256, q):
```

Thus for $n \geq 2$, $a(n) = u\,T^{n-2}\,v$ where u and v are row and column vectors respectively with $u_i = 1$ for i corresponding to configurations with no entry in the bottom row less than two of its three neighbours, 0 otherwise, and $v_i = 1$ for i corresponding to configurations with no entry in the top row less than two if its three neighbours, 0 otherwise. The following Maple code produces these vectors.

```
> qv:= proc(a) local r,s;
    s:= convert(16+floor((a-1)/16),base,4);
    r:= convert(16+((a-1) mod 16), base, 4);
    if nops(select(x -> x > s[1], [s[2],r[1],r[2]])) >= 2 then
  return 0 fi;
    if nops(select(x -> x > s[2], [s[1],r[1],r[2]])) >= 2 then
  return 0 fi;
    1
  end proc:
  v:= Vector(256,qv):
  qu:= proc(a) local r,s;
    s:= convert(16+floor((a-1)/16),base,4);
    r:= convert(16+((a-1) mod 16), base, 4);
    if nops(select(x -> x > r[1], [r[2],s[1],s[2]])) >= 2 then
```

```
    return 0 fi;
      if nops(select(x -> x > r[2], [r[1],s[1],s[2]])) >= 2 then
    return 0 fi;
      1
    end proc:
    u:= Vector[row](256,qu):
```

To check, here are the entries $a(2)$ to $a(10)$ of our sequence.

```
> seq(u . T^n . v, n = 0 .. 8);
```

$$28, 124, 602, 2776, 12922, 60720, 286047, 1335296, 6256326 \quad (1)$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \to t^{18} - 4\,t^{17} - 6\,t^{16} - 38\,t^{15} + 161\,t^{14} + 292\,t^{13} + 953\,t^{12} - 1185\,t^{11} - 4928\,t^{10} - 12336\,t^9 - 7076\,t^8 + 7076\,t^7 + 24464\,t^6 + 24608\,t^5 + 18624\,t^4 + 6720\,t^3 + 2304\,t^2 \quad (2)$$

This has degree 18, but is divisible by t^2. Thus we will have $0 = u\,P(T)\,T^{m-2}\,v = \sum_{i=2}^{18} p_i\,a(i+m)$ for $m \geq 2$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to the "empirical" homogeneous linear recurrence of order 16, with $n = m + 18 \geq 20$. It turns out that this is also true for $n \geq 17$.